



浙江省高等教育重点建设教材



单片机 实验与课程设计指导 (Proteus仿真版)

第二版

楼然苗 胡佳文 李光飞 编著
刘玉良 李韵磊 叶继英



ZHEJIANG UNIVERSITY PRESS
浙江大学出版社

浙江省高等教育重点建设教材

单片机实验与课程设计指导

(Proteus 仿真版)

(第二版)

楼然苗 胡佳文 李光飞 编著
刘玉良 李韵磊 叶继英



ZHEJIANG UNIVERSITY PRESS
浙江大学出版社

内容简介

单片机实验与课程设计指导是学生加深理论知识理解、提高实际设计能力的重要环节,从学生自己设计电路板,到程序编制与调试,最后完成一个单片机系统的设计,可以让学生体验成功的快乐!proteus 虚拟单片机仿真软件可以成功地进行绝大部分的单片机硬件仿真,从而在教师进行课堂教学或实验设计、演示等环节,可以轻松实现程序功能的展示。本书在原《单片机实验与课程设计指导(proteus 仿真版)》教材的基础上增加了新的实验与课程设计内容,适合教师在单片机课程教学中进行教学程序功能演示及作为学生实验与课程设计的指导用书。

图书在版编目(CIP)数据

单片机实验与课程设计指导:Proteus 仿真版 / 楼然苗等编著. —2 版. —杭州:浙江大学出版社, 2013. 7
ISBN 978-7-308-11906-1

I. ①单… II. ①楼… III. ①单片微型计算机—系统设计—应用软件—高等学校—教学参考资料 IV. ①TP368.1

中国版本图书馆 CIP 数据核字 (2013) 第 170919 号

单片机实验与课程设计指导(Proteus 仿真版)(第二版)

楼然苗 胡佳文 李光飞 编著
刘玉良 李韵磊 叶继英

责任编辑 吴昌雷
封面设计 刘依群
出版发行 浙江大学出版社
(杭州市天目山路 148 号 邮政编码 310007)
(网址: <http://www.zjupress.com>)
排版 杭州中大图文设计有限公司
印刷 杭州日报报业集团盛元印务有限公司
开本 787mm×1092mm 1/16
印张 15.75
字数 383 千
版印次 2013 年 7 月第 2 版 2013 年 7 月第 3 次印刷
书号 ISBN 978-7-308-11906-1
定价 32.00 元

版权所有 翻印必究 印装差错 负责调换

浙江大学出版社发行部联系方式:0571-88925591; <http://zjdxcbbs.tmall.com>

前 言

Proteus 是一种功能强大的电子设计自动化软件,提供智能原理图设计系统,能模拟数字电路、模拟电路及 MCU 器件混合仿真系统和 PCB 设计系统功能。Proteus 软件不仅可以仿真传统的电路分析实验、模拟电子线路实验、数字电路实验,而且可以提供嵌入式系统(单片机应用系统、ARM 应用系统)的仿真实验。它支持单片机和周边设备,可以仿真 51 系列、AVR、PIC、Motorola 的 68 系列等常用的 MCU,并可提供周边设备的仿真,例如字符 LCD 模块、图形 LCD 模块、LED 点阵、LED 七段显示模块、键盘/按键、直流/步进/伺服电机、RS232 虚拟终端、电子温度计、示波器等。Proteus 提供了大量的元件库,有 RAM、ROM、键盘、马达、LED、LCD、AD/DA、部分 SPI 器件、部分 IIC 器件等。在编译方面,它支持单片机汇编语言的编辑/编译/源码级仿真,也支持 Keil 和 MPLAB 等多种编译器。内带的 8051、AVR、PIC 的汇编编译器,也可以与第三方集成编译环境(如 IAR、Keil 和 Hitech)结合,进行高级语言的源码级仿真和调试。

利用 Proteus 的单片机硬件电路进行程序运行效果仿真,可以方便直观地进行单片机程序运行效果演示,极大地拓展了课堂教学及实验教学的硬件环境条件,老师或学生可以在教室或寝室方便地利用电脑进行单片机程序的调试及效果演示,为设计开发单片机应用产品提高了效率。

本书是用于教师或学生进行单片机实验或课程设计的指导书。本书中的所有实验程序及设计硬件电路资料可在浙江海洋学院精品课程网站(<http://61.153.216.116/jpkc/jpkc/dpj/News/Show.asp?id=60&cataid=A0049>)和浙江大学出版社网站(www.zjupress.com)中获得,以方便学校老师及学生教学与学习使用。用于 Proteus 仿真的单片机程序都可以在真实硬件电路板上运行,为教师课堂教学实验的演示或设计程序的功能演示提供了极大的

方便。

本书选用 89C52 系列单片机作为处理器,选择了课堂教学或课程实验中,以定时器使用、中断使用、串行口使用、七段 LED 显示器动态扫描显示、LCD 点阵液晶显示,以及在课程设计中彩灯控制器设计、单片机时钟设计、DS18B20 数字温度计设计、DS1302 实时时钟设计、低频信号发生器设计、电子密码锁等例子,较详细地介绍了系统功能、设计方案、硬件仿真电路、程序设计、仿真运行结果等,书中的源程序及电路图可供参考。

本书 Proteus 仿真电路图设计采用 Proteus 7.1 版本,使用时请安装 Proteus 7.1 及以上版本仿真软件。有关 Proteus 仿真软件的安装与使用方法请参考相关资料,本书中不进行相关介绍。

感谢浙江大学出版社在本书出版过程中给予的帮助与支持!

作者邮箱:louranmiao@zjou.edu.cn

作 者

2013 年 3 月

目 录

第 1 章 实验一:LED 小灯实验	(1)
1.1 实验内容与要求	(1)
1.2 实验一参考汇编程序	(2)
1.3 实验一参考 C 程序	(8)
第 2 章 实验二:定时/计数器实验	(13)
2.1 实验内容与要求	(13)
2.2 实验二参考汇编程序	(14)
2.3 实验二参考 C 程序	(19)
第 3 章 实验三:定时器中断实验	(23)
3.1 实验内容与要求	(23)
3.2 实验三参考汇编程序	(24)
3.3 实验三参考 C 程序	(26)
第 4 章 实验四:串行口通信实验	(30)
4.1 实验内容与要求	(30)
4.2 实验四参考汇编程序	(32)
4.3 实验四参考 C 程序	(37)
第 5 章 实验五:按键接口实验	(40)
5.1 实验内容与要求	(40)
5.2 实验五参考汇编程序	(42)
5.3 实验五参考 C 程序	(49)
第 6 章 实验六:八位共阳 LED 数码管实验	(53)
6.1 实验内容与要求	(53)

6.2	实验六参考汇编程序	(54)
6.3	实验六参考 C 程序	(62)
第 7 章	实验七: LCD 液晶显示器实验	(67)
7.1	实验内容与方法	(67)
7.2	实验七参考 C 程序	(68)
第 8 章	课程设计一:彩灯控制器	(99)
8.1	系统功能	(99)
8.2	设计方案	(99)
8.3	系统硬件仿真电路	(100)
8.4	程序设计	(100)
8.5	软件调试与运行结果	(101)
8.6	源程序清单	(102)
8.6.1	课程设计一参考汇编程序	(102)
8.6.2	课程设计一参考 C 程序	(106)
第 9 章	课程设计二:单片机时钟	(108)
9.1	系统功能	(108)
9.2	设计方案	(108)
9.3	系统硬件仿真电路	(109)
9.4	程序设计	(110)
9.4.1	主程序	(110)
9.4.2	显示子程序	(110)
9.4.3	定时器 T0 中断服务程序	(111)
9.4.4	T1 中断服务程序	(111)
9.4.5	调时功能程序	(111)
9.4.6	秒表功能程序	(111)
9.4.7	闹钟时间设定功能程序	(112)
9.5	软件调试与运行结果	(112)
9.6	源程序清单	(112)
9.6.1	课程设计二参考汇编程序	(112)
9.6.2	课程设计二参考 C 程序	(137)

第 10 章 课程设计三:DS1302 实时时钟	(142)
10.1 系统功能	(142)
10.2 设计方案	(142)
10.3 系统硬件仿真电路	(143)
10.4 程序设计	(143)
10.4.1 时钟读出程序设计	(143)
10.4.2 时间调整程序设计	(144)
10.5 软件调试与运行结果	(144)
10.6 源程序清单	(145)
10.6.1 课程设计三参考汇编程序	(145)
10.6.2 课程设计三参考 C 程序	(155)
第 11 章 课程设计四:数字温度计	(162)
11.1 系统功能	(162)
11.2 设计方案	(162)
11.3 系统硬件仿真电路	(163)
11.4 程序设计	(163)
11.4.1 主程序	(163)
11.4.2 读出温度子程序	(164)
11.4.3 温度转换命令子程序	(164)
11.4.4 计算温度子程序	(164)
11.4.5 显示数据刷新子程序	(165)
11.4.6 DS18B20 中的 ROM 命令	(165)
11.4.7 温度数据的计算处理方法	(167)
11.5 软件调试与运行结果	(168)
11.6 源程序清单	(168)
11.6.1 课程设计四参考汇编程序	(168)
11.6.2 课程设计四参考 C 程序	(178)
第 12 章 课程设计五:低频信号发生器	(184)
12.1 系统功能	(184)
12.2 设计方案	(184)

12.3	系统硬件仿真电路	(185)
12.3.1	控制部分	(185)
12.3.2	数模(D/A)转换部分	(185)
12.4	程序设计	(185)
12.5	软件调试与运行结果	(187)
12.6	源程序清单	(188)
12.6.1	课程设计五参考汇编程序	(188)
12.6.2	课程设计五参考 C 程序	(193)
第 13 章	课程设计六:16 点阵 LED 显示器	(196)
13.1	系统功能	(196)
13.2	设计方案	(196)
13.3	系统硬件仿真电路	(197)
13.3.1	单片机系统与外围电路	(197)
13.3.2	行驱动电路	(199)
13.3.3	列驱动电路	(199)
13.4	程序设计	(199)
13.4.1	显示驱动程序	(199)
13.4.2	系统主程序	(199)
13.5	软件调试与运行结果	(200)
13.6	源程序清单	(201)
13.6.1	课程设计六参考汇编程序	(201)
13.6.2	课程设计六参考 C 程序	(208)
第 14 章	课程设计七:电子密码锁	(213)
14.1	系统功能	(213)
14.2	设计方案	(213)
14.3	系统硬件仿真电路	(214)
14.3.1	单片机电路	(214)
14.3.2	键盘电路	(215)
14.3.3	数码管显示电路	(215)
14.3.4	密码存储电路	(215)
14.3.5	声光提示电路	(216)

14.4 程序设计.....	(216)
14.4.1 主程序.....	(216)
14.4.2 初始化程序.....	(217)
14.4.3 按键扫描程序.....	(217)
14.5 软件调试与运行结果.....	(218)
14.6 C 源程序清单.....	(220)
附录 1 51 系列单片机的特殊功能寄存器表.....	(233)
附录 2 51 系列单片机中断入口地址表.....	(235)
附录 3 51 系列单片机汇编指令表.....	(236)
参考文献.....	(242)

第 1 章

实验一:LED 小灯实验

1.1 实验内容与要求

1. 实验目的

- (1) 学习用程序延时的方法进行 LED 小灯的亮灭控制。
- (2) 学习掌握流水小灯的编程方法。

2. Proteus 仿真实验硬件电路

LED 小灯实验的 Proteus 仿真实验硬件电路如图 1.1 所示。




图 1.1 LED 小灯实验的仿真硬件电路

3. 实验任务

完成对接在 P1、P3 端口的发光二极管闪亮控制程序的设计和调试。具体要求：

- (1) 用程序延时的方法让 P1 口的 1 个 LED 小灯每隔 1s 交替闪亮。

- (2)用程序延时的方法让 P1 口的 8 个 LED 发光二极管循环闪亮(每个亮 50ms)。
- (3)用程序延时的方法让 P1 口的 8 个 LED 小灯追逐闪亮(50ms 间隔变化)。
- (4)用程序延时的方法让 P1、P3 口的 16 个 LED 小灯循环闪亮(每个亮 50ms)。

4. 实验预习要求

- (1)根据硬件电路原理图,分析二极管点亮的条件;复习延时子程序中延时时间的计算方法,会计算延时子程序的初值。
- (2)根据硬件电路原理图,画出实际接线图。
- (3)根据实验任务设计出相应的调试程序。
- (4)学习掌握 Wave、Madwin、Keil-51 等编译软件的使用方法。
- (5)完成预习报告。

5. 实验设备

计算机(已安装单片机汇编编译软件及 Proteus 软件)。

6. 实验报告要求

整理好实验任务 1~4 中经 Proteus 运行正确的程序。

1.2 实验一参考汇编程序

```

;*****;
;                                     实验程序 1.1                                     ;
;   用程序延时的方法让 P1 口的 1 个 LED 小灯每隔 1s 交替闪亮   ;
;                                     12MHz 晶振                                     ;
;*****;
ORG    0000H    ; 程序执行开始地址
LJMP   START    ; 跳至 START 执行
;
ORG    0030H    ; 以下程序放在 0030H 地址后
START: CPL      P1.0
        LCALL   DL1S
        AJMP    START
;
; 约 0.5ms 延时子程序,执行一次时间为 503  $\mu$ s
DL503: MOV     R2, #250
LOOP1: DJNZ    R2, LOOP1

```

```
RET
;
;约 10ms 延时子程序(调用 20 次 0.5ms 延时子程序)
DL10ms: MOV    R3, # 20
LOOP2:  LCALL   DL503
        DJNZ    R3, LOOP2
RET
;
;约 1s 延时子程序
DL1S:   MOV     R4, # 100
LOOP3:  LCALL   DL10ms
        DJNZ    R4, LOOP3
RET
;
END      ;  结束
```

实验程序 1.1 的 Proteus 仿真效果如图 1.2 所示。




图 1.2 实验程序 1.1 的 Proteus 仿真效果

```

;*****;
;                               实验程序 1.2                               ;
;用程序延时的方法让 P1 口的 8 个 LED 发光二极管循环闪亮(每个亮 50ms);
;                               12MHz 晶振                               ;
;*****;

    ORG    0000H    ; 程序执行开始地址
    LJMP   START    ; 跳至 START 执行

;

    ORG    0030H    ; 以下程序放在 0030H 地址后
START: LCALL  FUN0
    AJMP   START

;
;循环闪亮功能子程序
FUN0:  MOV    A, #0FEH    ; 累加器赋初值
FUN00: MOV    P1, A        ; 累加器值送至 P1 口
        LCALL  DL50ms      ; 延时
        JNB    ACC.7, OUT   ; 累加器最高位为 0 时结束
        RL     A            ; 累加器 A 中数据循环左移 1 位
        AJMP   FUN00        ; 转 FUN00 循环
OUT:    RET

;
;约 0.5ms 延时子程序,执行一次时间为 503  $\mu$ s
DL503: MOV    R2, #250
LOOP1: DJNZ    R2, LOOP1
        RET

;
;约 50ms 延时子程序(调用 100 次 0.5ms 延时子程序)
DL50ms: MOV    R3, #100
LOOP2: LCALL  DL503
        DJNZ    R3, LOOP2
        RET

;
END

```

实验程序 1.2 的 Proteus 仿真效果如图 1.3 所示。




图 1.3 实验程序 1.2 的 Proteus 仿真效果

```

;*****
;
;          实验程序 1.3
;
; 用程序延时的方法让 P1 口的 8 个 LED 小灯追逐闪亮(50ms 间隔变化);
;
;          12MHz 晶振
;
;*****
ORG      0000H      ; 程序执行开始地址
LJMP     START      ; 跳至 START 执行
;
ORG      0030H      ; 以下程序放在 0030H 地址后
START:  LCALL  FUN1
        AJMP   START
;
;追逐闪亮功能子程序
FUN1:   MOV    A, # 0FEH      ; 累加器赋初值
FUN11:  MOV    P1, A          ; 累加器值送至 P1 口
        LCALL  DL50ms        ; 延时
        JZ     OUT            ; A 为 0 结束
        RL     A              ; 累加器 A 中数据循环左移 1 位

```

```

ANL    A,P1          ;A 同 P1 口值相与
AJMP   FUN11         ;转 FUN11 循环

OUT:   RET

;

;约 0.5ms 延时子程序,执行一次时间为 503  $\mu$ s
DL503: MOV    R2, #250
LOOP1: DJNZ   R2, LOOP1
      RET

;约 50ms 延时子程序(调用 100 次 0.5ms 延时子程序)
DL50ms: MOV   R3, #100
LOOP2: LCALL  DL503
      DJNZ   R3, LOOP2
      RET

;
END

```

实验程序 1.3 的 Proteus 仿真效果如图 1.4 所示。




图 1.4 实验程序 1.3 的 Proteus 仿真效果

```

;*****
;                                     实验程序 1.4
; 用程序延时的方法让 P1、P3 口的 16 个 LED 小灯循环闪亮(每个亮 50ms) ;
;                                     12MHz 晶振
;*****
        ORG    0000H    ; 程序执行开始地址
        LJMP   START    ; 跳至 START 执行
;
        ORG    0030H    ; 以下程序放在 0030H 地址后
START:  LCALL  FUNP1
        MOV    P1, #0FFH
        LCALL  FUNP3
        MOV    P3, #0FFH
        AJMP   START
;
; 循环闪亮功能子程序
FUNP1:  MOV    A, #0FEH    ; 累加器赋初值
FUN11:  MOV    P1, A        ; 累加器值送至 P1 口
        LCALL  DL50ms      ; 延时
        JNB    ACC. 7, OUT  ; 累加器最高位为 0 时结束
        RL     A            ; 累加器 A 中数据循环左移 1 位
        AJMP   FUN11       ; 转 FUN11 循环
OUT:    RET
FUNP3:  MOV    A, #0FEH    ; 累加器赋初值
FUN33:  MOV    P3, A        ; 累加器值送至 P3 口
        LCALL  DL50ms      ; 延时
        JNB    ACC. 7, OUT  ; 累加器最高位为 0 时结束
        RL     A            ; 累加器 A 中数据循环左移 1 位
        AJMP   FUN33       ; 转 FUN33 循环
;
; 约 0.5ms 延时子程序, 执行一次时间为 503  $\mu$ s
DL503:  MOV    R2, #250
LOOP1:  DJNZ   R2, LOOP1
        RET
;
; 约 50ms 延时子程序(调用 100 次 0.5ms 延时子程序)
DL50ms: MOV    R3, #100
LOOP2:  LCALL  DL503
        DJNZ   R3, LOOP2
        RET
;
END

```